SIMPLICIAL COMPLEXES IN MACAULAY2

BEN HERSEY, GREGORY G. SMITH, AND ALEXANDRE ZOTINE

ABSTRACT. We highlight some features of the *SimplicialComplexes* package in *Macaulay2*.

This updated version of the *SimplicialComplexes* package in *Macaulay2* [M2], originally developed by Sorin Popescu, Gregory G. Smith, and Mike Stillman, adds constructors for many classic examples, implements a new data type for simplicial maps, and incorporates many improvements to the methods and documentation. Emphasizing combinatorial and algebraic applications, the primary data type encodes an abstract simplicial complex—a family of subsets that is closed under taking subsets. These simplicial complexes should not be conflated with their geometric realizations formed from points, line segments, filled-in triangles, solid tetrahedra, and their higher-dimensional analogues in some Euclidean space. The subsets in a simplicial complex are called its faces, the faces having cardinality 1 are its vertices, and the maximal faces (ordered by inclusion) are its facets. Following the combinatorial conventions, every nonempty simplicial complex has the empty set as a face.

In this package, a simplicial complex is represented by its Stanley–Reisner ideal. The vertices are identified with a subset of the variables in a polynomial ring and each face is identified with the product of the corresponding variables. A nonface is any subset of the variables that does not belong to the simplicial complex and each nonface is again identified with a product of variables. The Stanley–Reisner ideal of a simplicial complex is generated by monomials corresponding to its nonfaces; see Definition 5.1.2 in [BH], Definition 1.6 in [MS], or Definition II.1.1 in [S]. Because computations in the associated polynomial ring are typically prohibitive, this package is not intended for simplicial complexes with a large number of vertices.

CONSTRUCTORS. The basic constructor for a simplicial complex accepts two different kinds of input. Given a list of monomials, it returns the smallest simplicial complex containing the corresponding faces. Given a radical monomial ideal, it returns the simplicial complex having the input as its Stanley–Reisner ideal. We illustrate both methods using the 'bowtie' complex appearing in Figure 1.

```
Macaulay2, version 1.18
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,
               LLLBases, MinimalPrimes, PrimaryDecomposition, ReesAlgebra,
               Saturation, TangentCone

i1 : needsPackage "SimplicialComplexes";

i2 : S = QQ[x_0..x_4];

i3 : ⋈ = simplicialComplex {x_0*x_1*x_2, x_2*x_3*x_4}

o3 = simplicialComplex | x_2x_3x_4 x_0x_1x_2 |

o3 : SimplicialComplex

i4 : I = monomialIdeal ⋈

o4 = monomialIdeal (x x , x x , x x , x x )
                     0 3   1 3   0 4   1 4

o4 : MonomialIdeal of S
```

2020 *Mathematics Subject Classification*. 05E45, 13F55, 55U10

`SimplicialComplexes` version 2.0 .

FIGURE 1. The left is the bowtie complex ⋈ and the right its Alexander dual ⋈*

```
i5 : ⋈' = simplicialComplex I

o5 = simplicialComplex | x_2x_3x_4 x_0x_1x_2 |

o5 : SimplicialComplex

i6 : assert(⋈ === ⋈')
```

The package also has convenience constructors for some archetypal simplicial complexes. For example, we recognize triangulations of the real projective plane and the Klein bottle from their reduced homology groups; see Theorems 6.3–6.4 in [Mu].

```
i7 : ℙ = realProjectiveSpaceComplex(2, R = ZZ[a..h])

o7 = simplicialComplex | bef aef cdf adf bcf cde bde ace abd abc |

o7 : SimplicialComplex

i8 : for j from 0 to 2 list prune HH_j ℙ

o8 = {0, cokernel | 2 |, 0}

o8 : List

i9 : for j from 0 to 2 list prune HH_j kleinBottleComplex R

o9 = {0, cokernel | 2 |, 0}
                  | 0 |
o9 : List
```

More comprehensively, Frank H. Lutz enumerates simplicial complexes having a small number of vertices; see [L]. Using this list, the package creates a database of 43 138 simplicial 2-manifolds having at most 10 vertices and 1 343 simplicial 3-manifolds having at most 9 vertices. We demonstrate this feature by exhibiting the distribution of f-vectors among the 3-manifolds having 9 vertices. For all nonnegative integers j, the j-th entry in the f-vector is the number of faces having j vertices.

```
i10 : tally for j from 0 to 1296 list
          fVector smallManifold(3, 9, j, ZZ[vars(1..9)])

o10 = Tally{{1, 9, 26, 34, 17} => 7  }
            {1, 9, 27, 36, 18} => 23
            {1, 9, 28, 38, 19} => 45
            {1, 9, 29, 40, 20} => 84
            {1, 9, 30, 42, 21} => 128
            {1, 9, 31, 44, 22} => 175
            {1, 9, 32, 46, 23} => 223
            {1, 9, 33, 48, 24} => 231
            {1, 9, 34, 50, 25} => 209
            {1, 9, 35, 52, 26} => 121
            {1, 9, 36, 54, 27} => 51
o10 : Tally
```

Exploiting the same loop, we construct the simplicial maps from a minimal triangulation of a torus to the induced subcomplex on the first 7 vertices for each of these 3-manifolds.

```
i11 : T = smallManifold(2, 7, 1, R = ZZ[a..i])

o11 = simplicialComplex | dfg afg deg aeg adf cde ace bcd abd abc |

o11 : SimplicialComplex

i12 : gens monomialIdeal T

o12 = | acd be ade bf cf ef bg cg adg h i |

               1       11
o12 : Matrix R  <--- R

i13 : for j from 0 to 2 list prune HH_j T

                1
o13 = {0, 0, ZZ }

o13 : List

i14 : mapList = for j from 0 to 1296 list (
          phi := map(smallManifold(3, 9, j, S), T, gens R);
          if not isWellDefined phi then continue else phi);

i15 : # mapList

o15 = 319

i16 : assert all(mapList, phi -> isInjective phi)
```

COMBINATORIAL TOPOLOGY. We showcase some of the key operations on simplicial complex using the bowtie complex. Viewing a simplicial complex as lying in a standard simplex yields a duality theory. For any simplicial complex Δ whose vertices belong to a set V, the Alexander dual is the simplicial complex $\Delta^* := \{F \subseteq V \mid V \setminus F \in \Delta\}$. Since each simplicial complex in this package has an underlying polynomial ring, the variables in this ring form the canonical superset of vertices.

```
i17 : dual ⋈

o17 = simplicialComplex | x_1x_2x_4 x_0x_2x_4 x_1x_2x_3 x_0x_2x_3 |

o17 : SimplicialComplex

i18 : assert(dual dual ⋈ === ⋈)

i19 : assert(dual monomialIdeal ⋈ === monomialIdeal dual ⋈)
```

Algebraically, Alexander duality switches the roles of minimal generators and irreducible components in the Stanley–Reisner ideal.

```
i20 : monomialIdeal dual ⋈

o20 = monomialIdeal (x x , x x )
                      0 1   3 4

o20 : MonomialIdeal of S

i21 : irreducibleDecomposition monomialIdeal ⋈

o21 = {monomialIdeal (x , x ), monomialIdeal (x , x )}
                       0   1                   3   4

o21 : List
```

The topological form of Alexander duality gives an isomorphism between the reduced homology of a simplicial complex and reduced cohomology of its dual; see Theorem 5.6 in [MS].

```
i22 : n = numgens ring ⋈;
```

```
i23 : assert all(-1..n-1, j -> prune HH^(n-j-3) dual ⧓ == prune HH_j ⧓)
```

A simplicial complex Δ is Cohen–Macaulay if the associated quotient ring S/I, where I is the Stanley–Reisner ideal of Δ in the polynomial ring S, is Cohen–Macaulay. To characterize this attribute topologically, we introduce a family of subcomplexes. For any face F in Δ, the link is the subcomplex $\operatorname{link}_\Delta(F) := \{G \in \Delta \mid F \cup G \in \Delta \text{ and } F \cap G = \varnothing\}$. The link of the vertex x_2 in ⧓ has two disjoint facets.

```
i24 : L = link(⧓, x_2)

o24 = simplicialComplex | x_3x_4 x_0x_1 |

o24 : SimplicialComplex

i25 : prune HH_0 L

        1
o25 = QQ

o25 : QQ-module, free
```

The dimension of a simplicial complex is one less than the maximal cardinality of its faces.

```
i26 : dim L

o26 = 1
```

As discovered by Gerald Reisner, the simplicial complex Δ is Cohen–Macaulay if and only if, for all faces F in Δ and all integers j less than the dimension of $\operatorname{link}_\Delta(F)$, the j-th reduced homology group of $\operatorname{link}_\Delta(F)$ vanishes; see Corollary 5.3.9 in [BH], Theorem 5.53 in [MS], or Corollary II.4.2 in [S]. Using this criterion, the 0-th reduced homology certifies that ⧓ is not Cohen–Macaulay.

```
i27 : assert(HH_0 L != 0)

i28 : assert(dim(S^1/monomialIdeal ⧓) =!= n - pdim(S^1/monomialIdeal ⧓))
```

However, the 1-skeleton of ⧓ is Cohen–Macaulay.

```
i29 : ⋈ = skeleton(1, ⧓)

o29 = simplicialComplex | x_3x_4 x_2x_4 x_2x_3 x_1x_2 x_0x_2 x_0x_1 |

o29 : SimplicialComplex

i30 : faceList = rsort flatten values faces ⋈

o30 = {x x , x x , x x , x x , x x , x x , x , x , x , x , x , 1}
        0 1   0 2   1 2   2 3   2 4   3 4   0   1   2   3   4

o30 : List

i31 : assert all(faceList, F -> (L := link(⋈, F); all(dim L, j -> HH_j L == 0)))

i32 : assert(dim(S^1/monomialIdeal ⋈) === n - pdim(S^1/monomialIdeal ⋈))
```

Alternatively, we verify that bowtie complex ⧓ is not Cohen–Macaulay by showing that its h-vector has a negative entry; see Theorem 5.1.10 in [BH] or Corollary II.2.5 in [S]. By definition, the h-vector of a simplicial complex Δ is a binomial transform of its f-vector: for all $0 \leqslant j \leqslant d := \dim \Delta$, we have $h_j = \sum_{k=0}^{j} (-1)^{j-1} \binom{d+1-k}{j-k} f_{k-1}$. The h-vector encodes the numerator of the Hilbert series for S/I.

```
i33 : d = dim ⧓

o33 = 2
```

```
i34 : faces ⋈

o34 = HashTable{-1 => {1}                                }
                0 => {x , x , x , x , x }
                       0   1   2   3   4
                1 => {x x , x x , x x , x x , x x , x x }
                       0 1   0 2   1 2   2 3   2 4   3 4
                2 => {x x x , x x x }
                       0 1 2   2 3 4

o34 : HashTable

i35 : fVec = fVector ⋈

o35 = {1, 5, 6, 2}

o35 : List

i36 : hVec = for j from 0 to d list
           sum(j+1, k -> (-1)^(j-k) * binomial(d+1-k, j-k) * fVec#k)

o36 = {1, 2, -1}

o36 : List

i37 : hilbertSeries(S^1/monomialIdeal ⋈, Reduce => true)

                2
      1 + 2T - T
o37 = -----------
              3
        (1 - T)

o37 : Expression of class Divide
```

RESOLUTIONS OF MONOMIAL IDEALS. In this section, we will overview some chain complex constructions arising from simplicial complexes. We give an example of ideal homogenization and resolutions supported on a simplicial complex, and we provide examples of Taylor, Lyubeznik, and Buchberger resolutions; as well as Scarf complexes.

Let Δ be a simplicial complex with q vertices and let $\widetilde{C}(\Delta;k)$ be the augmented chain complex of Δ with incidence function ε. We will let $S = k[x_0,\dots,x_n]$ be a polynomial ring over the commutative ring k, which is typically a field. We assume that the ring S has the fine $\mathbb{N}^{n+1}$ grading.

For a monomial ideal $I \subset S$, minimally generated by $m_1, m_2, \dots, m_q$, a **labelling** of Δ by I is a bijection which assigns a minimal generator of I to each vertex of Δ. Without loss of generality, we will assume this assignment is $v_i \longmapsto m_i$ for $i = 1,2,\dots,q$. For each face $F \in \Delta$ we can construct the monomial $m_F = \operatorname{lcm}(m_i \mid i \in F)$. The I-**homogenization** of $\widetilde{C}(\Delta;k)$ is the chain complex $(\mathbf{G}, d)$ such that

$$G_0 = S \;\text{ and }\; G_i = \bigoplus_{\dim(F)=i-1} S(m_F).$$

We will use $\{f_F \mid F \in \Delta\}$ to denote the canonical basis of G and we define the differential of $\mathbf{G}$ using

$$d(f_F) \;=\; \sum_{\dim(F)=i-1} \frac{m_F}{m_{F'}} \cdot \varepsilon(F,F') f_{F'}.$$

We say that I has a **resolution supported on** Δ if the I-homogenization of $\widetilde{C}(\Delta;k)$ is a free resolution for some labelling of the vertices of Δ. For further details on I-homogenization, we refer the reader to any one of [MS, P].

Let $S = \mathbb{Q}[x_0, x_1, x_2, x_3]$, let $I = (x_0x_1, x_0x_2, x_0x_3, x_1x_2x_3)$, and let Γ be the simplicial complex with facets $\{v_1, v_2, v_3\}$ and $\{v_2, v_4\}$. If we use the labelling $v_1 \longmapsto x_0x_1$, $v_2 \longmapsto x_0x_2$, $v_3 \longmapsto x_0x_3$, and $v_4 \longmapsto x_1x_2x_3$, we get the labelled simplicial complex shown in Figure 2.



FIGURE 2. The homogenization of Γ

We can use Macaulay2 to compute both $\widetilde{C}(\Delta;\mathbb{Q})$ and the I-homogenization of Δ relative to this labelling, which is the minimal free resolution of S/I.

```
i9 : R = ZZ/101[y_0..y_13];
i10 : S = QQ[x_0..x_3];
i11 : Δ = simplicialComplex{R_0*R_1*R_2, R_2*R_3};
i12 : I = ideal(x_0*x_1,x_0*x_2,x_0*x_3,x_1*x_2*x_3);
o12 : Ideal of S
i13 : C = chainComplex Δ
        ZZ 1       ZZ 4       ZZ 4       ZZ 1
o13 = (---)  <-- (---)  <-- (---)  <-- (---)
       101        101        101        101
      -1         0          1          2
o13 : ChainComplex
i24 : G = chainComplex(Δ, Labels => {x_0*x_1,x_0*x_2,x_0*x_3,x_1*x_2*x_3});
i25 : G.dd
           1                                         4
o25 = 0 : S  <-------------------------------------- S  : 1
              | x_0x_1 x_0x_2 x_0x_3 x_1x_2x_3 |
           4                                     4
      1 : S  <---------------------------------- S  : 2
              {2} | -x_2 -x_3 0    0       |
              {2} | x_1  0    -x_3 0       |
              {2} | 0    x_1  x_2  -x_1x_2 |
              {3} | 0    0    0    x_0     |
           4                  1
      2 : S  <---------------- S  : 3
              {3} | x_3  |
              {3} | -x_2 |
              {3} | x_1  |
              {4} | 0    |
o25 : ChainComplexMap
i15 : (res (S^1/I)) == G
o15 = true
```

The I-homogenization of Δ is dependent on how you label the vertices, and that is reflected by the ordering of the monomials in the Labels argument. Indeed, if we swap the labels of v_1 and v_4, then the I-homogenization is no longer a resolution of S/I.

```
i22 : G' = chainComplex(Δ, Labels => {x_1*x_2*x_3,x_0*x_2,x_0*x_3,x_0*x_1});
```

```
i23 : prune homology G'
o23 = 0 : cokernel | x_0x_3 x_0x_2 x_0x_1 x_1x_2x_3 |
      1 : cokernel {3} | x_3 |
      2 : 0
      3 : 0
o23 : GradedModule
```

Given a monomial ideal I, there are several algorithms that will produce a simplicial complex Δ and a labelling of Δ by the minimal generators of I such that the I-homogenization of $\widetilde{C}(\Delta;k)$ is a free resolution of S/I, though often non-minimally. Examples of such constructions are the Taylor resolution, Lyubeznik resolution, and the Buchberger resolution, all of which are implemented in `SimplicialComplexes`. We have also implemented a constructor for the Scarf complex, which is a complex that is not always a free resolution of S/I, but when it is a free resolution it is minimal. We will not describe these constructions here, but a concise description of the Taylor resolution, Lyubeznik resolution, and Scarf complex is given in [Me], and a description of the Buchberger resolution is given in [OW].

Consider the monomial ideal $I = (x_1x_3, x_2^2, x_0x_2, x_1^2, x_0^2) \subset \mathbb{C}[x_0,\dots,x_3]$. The Taylor resolution of I can be realized as an I-homogenization of the 4-simplex.

```
i2 : R = QQ[a,b,c,d,e];
i3 : S = QQ[x_0..x_3];
i4 : I = monomialIdeal(x_1*x_3, x_2^2,  x_0*x_2, x_1^2, x_0^2);
o4 : MonomialIdeal of S
i5 : T = taylorResolution J
      1      5       10       10       5      1
o5 = S  <-- S  <-- S   <-- S   <-- S  <-- S
     0      1      2       3       4      5
o5 : ChainComplex
i6 : T == chainComplex(simplexComplex(4,R),Labels => first entries mingens I)
o6 = true
```

The Buchberger simplicial complex is a subcomplex of the 4-simplex, and the Buchberger resolution is an I-homogenization of the Buchberger simplicial complex. For this example, the Buchberger resolution is the minimal free resolution of S/I, but this is not always the case.

```
i7 : buchbergerSimplicialComplex(J,R)
o7 = simplicialComplex | acde abcd |
o7 : SimplicialComplex
i8 : B = buchbergerResolution J
      1      5      9      7      2
o8 = S  <-- S  <-- S  <-- S  <-- S
     0      1      2      3      4
o8 : ChainComplex
i10 : betti B === betti(res J)
o10 = true
```

Lyubeznik simplicial complexes and resolutions are constructed relative to a total order on the minimal generators of I. Every ordering will produce a resolution, but these resolutions need not be isomorphic. When no ordering is given, the methods `lyubeznikSimplicialComplex` and `lyubeznikResolution` will order the generators relative to the monomial order on S which, in Macaulay2, is graded revlex by default. The option `MonomialOrder` reorders the minimal generators of I relative to the monomial ordering on S. For example, `MonomialOrder => {2,1,0,3,4}` refers to the total ordering $x_0x_2 < x_2^2 < x_1x_3 < x_1^2 < x_0^2$ on the minimal generators of I. We see that by changing the ordering we can both produce the worst case (Taylor resolution) and best case (minimal free resolution).

```
i11 : lyubeznikSimplicialComplex(J,R)

o11 = simplicialComplex | abcde |

o11 : SimplicialComplex

i12 : lyubeznikResolution(J) == taylorResolution(J)

o12 = true

i13 : lyubeznikSimplicialComplex(J, R, MonomialOrder => {2,1,0,3,4})

o13 = simplicialComplex | acde abcd |

o13 : SimplicialComplex

i14 : L = lyubeznikResolution(J,MonomialOrder => {2,1,0,3,4})

       1      5      9      7      2
o14 = S  <-- S  <-- S  <-- S  <-- S
                                   
      0      1      2      3      4

o14 : ChainComplex
```

The Scarf simplicial complex of I starts with the labelled 4-simplex and removes any faces F, F' such that $m_F = m_{F'}$. The I-homogenization of the Scarf simplicial complex is the Scarf chain complex. It is often the case that the Scarf chain complex is not a free resolution of S/I, but when it is a resolution, it is minimal, see [?BPS, Lemma 3.1].

```
i16 : scarfSimplicialComplex(J,R)

o16 = simplicialComplex | acde abcd |

o16 : SimplicialComplex

i17 : scarfChainComplex J == buchbergerResolution J

o17 = true
```

ACKNOWLEDGEMENTS. All three authors were partially supported by the Natural Sciences and Engineering Research Council of Canada (NSERC).

REFERENCES

[ÀFG] Josep Àlvarez Montaner, Oscar Fernández-Ramos, and Philippe Gimenez, *Pruned cellular free resolutions of monomial ideals*, J. Algebra **541** (2020), 126–145.

[BH] Winfried Bruns and Jürgen Herzog, *Cohen-Macaulay rings*, Cambridge Studies in Advanced Mathematics, vol. 39, Cambridge University Press, Cambridge, 1993.

[L] Frank H. Lutz, *The Manifold Page* (2017), http://page.math.tu-berlin.de/~lutz/stellar/.

[M2] Daniel R. Grayson and Michael E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at http://www.math.uiuc.edu/Macaulay2/.

[Me] Jeff Mermin, *Three simplicial resolutions*, Progress in commutative algebra 1, de Gruyter, Berlin, 2012, pp. 127–141.

[MS] Ezra Miller and Bernd Sturmfels, *Combinatorial Commutative Algebra*, Graduate Texts in Mathematics, vol. 227, Springer-Verlag New York, 2005.

[Mu] James R. Munkres, *Elements of algebraic topology*, Addison-Wesley Publishing Company, Menlo Park, CA, 1984.

[P] Irena Peeva, *Graded Syzygies*, Algebra and Applications, vol. 14, Springer-Verlag London, 2011.

[OW] Anda Olteanu and Volkmar Welker, *The Buchberger resolution*, J. Commut. Algebra **8** (2016), no. 4, 571–587.

[S] Richard P. Stanley, *Combinatorics and Commutative Algebra*, 2nd ed., Progress in Mathematics, vol. 41, Birkhäuser Boston, 1996.

DEPARTMENT OF MATHEMATICS AND STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ONTARIO, K7L 3N6
hersey.b@queensu.ca, ggsmith@mast.queensu.ca, 18az45@queensu.ca.